

	cd $firmpath
	global firstyear 1996
	global lastyear 2006 
	global time_constraint  // we add a constraint to deal with left censoring 
	global time_constraint2 "drop if firstyear_sample>2004" // drop right-censored seller-buyer relationships
    global product hs4 
	global dim_decile $product  
	global variant _baseline
	
	use $correspath/corres_nc8$firstyear$lastyear, clear
	keep if year==2002
	rename nc8 nc8_2002
	tostring nc8_2002, replace 
	replace nc8_2002="0"+ nc8_2002 if length(nc8_2002)==7 
	g hs6_2002=substr(nc8_2002,1,6) 
	egen count=count(id_conc), by(id_conc)
	drop if count>30 // drop 2 id_conc that have more than 30 CN8 products (79 and 139 resp.) 
	keep id_conc hs6 
	duplicates drop id_conc hs6, force 
	save temp, replace

	use new_$dataset, clear 
	sort id_conc
	joinby id_conc using temp
	
	g hs4=substr(hs6,1,4)
	collapse (sum) export, by(iso2 year month buyer siren $product)
	destring hs4, replace 
	save temp_hs4, replace 
	
	global variant hs4
	construct_base_4decile temp_hs4 base_4decile$firstyear$lastyear$variant "$time_constraint" "$time_constraint2" whole "$countrysample" 

	use base_4decile$firstyear$lastyear$variant, clear
	sort iso2 id_bp id_bs id_bs_continuous
	id_group _ "iso2 $product id_bp" // id_group does the same job as egen group (but work if we have millions of groups)
	drop id_bp
	rename _ id_bp
	id_group _ "iso2 $product id_bs"
	drop id_bs
	rename _ id_bs
	id_group _ "iso2 $product id_bs_continuous"
	drop id_bs_continuous
	rename _ id_bs_continuous
	egen p=group($product)
	drop if export==0
	g dur=length_bps
	g size=export // used to be totx/dur, but this creates some issues as firms do not need inputs every month
	collapse (mean) size dur  , by(id_bs_continuous $product iso2) 
	g lsize=log(size)
	egen _=mean(lsize), by($product)
	g lsize_= size- _ 
	egen min=pctile(lsize_), by($dim_decile)  p(1) 
	egen max=pctile(lsize_), by($dim_decile)  p(99) 
	egen decile=xtile(lsize_) , by($dim_decile) nq(10) 
	drop if lsize_<min | lsize_>max 
	save base_4decile$firstyear$lastyear$variant, replace
	
	use base_4decile19962006$variant, clear 
	egen hs=group(hs4)
	local 2 dur 
    collapse (mean) size `2' (median) med_size=size med_`2'=`2', by($product iso2 decile hs) 
	replace `2'=log(`2')
	g fdecile=log(log(10 - decile) - log(9 - decile))
	replace fdecile=log(log(1.1)) if decile==1 // this is for size from pctile 0.01 to pctile 0.1 
	replace fdecile=log(log(10)) if decile==10 // this is for size from pctile 0.9 to pctile 0.99 
	replace `2'=`2' /0.9 if decile==1 |decile==10 // to account for the fact that we drop the top1% and bottom 1%
	egen p=group($product)
	egen c=group(iso2)
	reghdfe `2' fdecile , a(rs_hs4=hs) 
	collapse (mean) rs* `2', by(hs4)
	save $rspath/rs_$firstyear$lastyear$variant, replace // Relationship stickiness (rs) measures by product are in this dataset 
